Maple Overview

1. Starting Up and Shutting Down

Login to a computer in the Math Lab using your Whitman username and the Math Lab password. Once you log in, you'll see a screen that looks a lot like Windows.

At the bottom of the screen you'll see an icon that looks like a maple leaf with an integral sign on it. Click this once to bring up a Maple Worksheet.

To exit, choose FILE → EXIT, like you normally do in Windows.

Be sure to save your work! FILE → SAVE

When you're ready to leave the lab, be sure to LOG OUT:

- With your mouse, click on the big letter K (at the bottom of the screen, to your left).
- Select Logout (it's the last option).
- Select "Log in as different user"

2. Practice with Arithmetic

(1) PRACTICE SESSION: Type in the following, and note what comes out.

```
? sin
? numer
```

The question mark is how to get help on Maple commands. Here, we saw the help pages for the sine function and for the "numer" function.

```
2+4;
134^39;
3/5 + 5/9 + 7/12;
sqrt(24);
4*(3+Pi);
sin(5*Pi/3);
exp(x);
```

Here we see that Maple does exact arithmetic. Also, to input e^x, we had to use `exp(x)`

(2) PRACTICE SESSION: Type in the following, and note what comes out.

```
seq(k^2,k=1..100);
```

This command performs something like constructing a sequence from calculus. It performs the operation for $k = 1$, then for $k = 2$ and so on until $k = 100$.

```
ifactor(31722722304);
```

This gives you a prime factorization of a very large number!

```
evalf(3/5 + 5/9 + 7/12);
evalf(Pi,3);
evalf(Pi,100);
```

To get a numberical approximation, use `evalf`. To tell Maple how many digits, there is an optional second argument (e.g., 3 and 100 in the last two commands).

(3) PRACTICE SESSION: Type in the following, and note what comes out.

```
joe:=32;
sqrt(joe);
```

The important thing here is the use of `:=` This is an assignment operation, where the thing on the right is assigned to the name on the left. Here, the value 32 is stored in the variable "joe", so we can take its square root (and otherwise act on it like it was a number).

Maple differentiates between the assignment operation, `:=` and treats it differently than an equal sign, `=`

```
sqrt(32.0);
result:=seq(sqrt(k),k=1..10);
evalf(result);
```

(4) SPECIAL TOPIC: Clearing your variables and re-setting Maple.

Here's an example of first assigning a number to a variable, then clearing it:

```
x:=65;
w:=x^2-4*x+7;
x:='x';
w:=x^2-4*x+7;
```

To clear Maple's memory completely, type `restart` For example,

```
p:=4; h:=Pi;
p; x; h;
restart;
p; x; h;
```

3. Practice with Algebra

(1) Problem: Save the expression $3x^2 + 8$ into a variable W, then substitute $x = 4$ into the expression.

```
W:=3*x^2+8;
subs(x=4,W);
```

Substitute the expression $4u - 5$ in for x.

```
subs(x=4*u-5,W);
```

Do multiple substitutions or more complex substitutions:

```
U:=(2/5)*x^2+3*y;
subs(x=3, y=4, U);
U:=sin(x)/sqrt(1-sin(x));
subs(sin(x)=y,U);
```

(2) There is a difference between an *equation* and an *assignment*. The `:=` notation is to assign the left hand side to the variable on the right hand side. The following will give you an error:

```
A+B:=6;
```

But we can assign the equation $A + B = 6$ into a third variable:

```
eqn1:=A+B=6;
```

and make substitutions:

```
subs(A=3,B=2,eqn1);
```

(3) We can *expand*, *factor*, and *simplify* algebraic expressions:

```
H:=2*(x-2)*(2*x^2+5*x+2)*(x+4);
factor(H);
A := (x^3-7*x^2+15*x-9)/(x^2+4*x+4)
factor(A);
factor(3*x^4-2*x^3+22*x^2-18*x-45);
ww:=x^(1/2)-x^(3/2);
factor(ww);
h:=7/(x+2)+(3*x/(x+2)^2);
simplify(h);
h:=sin(3*t)-sin(7*t);
simplify(h);
```

4. Functions

In Maple, there is a difference between a *function* and an *expression*. Here is how we define a function:

```
f:=x->sin(x)/x;
f(3);
f(x+h);
```

Here is the same thing, but in the form of an expression:

```
G:=sin(x)/x;
subs(x=3, G);
subs(x=x+h, G);
```

So which should you use? Usually its easier to use an expression than a function, but these methods are equivalent.

5. Calculus

The main computations in Calculus are the limit, the derivative, and the integral. If you're continuing on from the previous section, type `restart;` before going on.

(1) Compute $\lim_{x\to 0} \dfrac{\sin(x)}{x}$

```
A:=sin(x)/x;
limit(A,x=0);
```

Compute $\lim_{x\to 0} \dfrac{1}{x}$

```
limit(1/x,x=0);
```

We can compute multivariate limits:

```
A:=(x^2-y^2)/(x^2+y^2);
limit(A,{x=0,y=0});
```

Or even limits that should probably not exist:

```
limit(sin(1/x),x=0);
limit(a*x,x=infinity);
```

(2) The basic derivative (using an *expression*):

```
A:=(x^2-y^2)/(x^2+y^2);
diff(A,x);
diff(A,y);
```

(3) Basic integration:

```
int(sin(x),x=0..3*Pi);
int(1/x^2,x=1..infinity);
int(x/(x^3-1),x);
int(exp(-x^2)*ln(x),x);
```

"Inert" integration: Will give a numerical approximation

```
L:=Int(exp(-x^2)*ln(x),x=1..3);
evalf(L);
```

We can compute improper integrals:

```
int(exp(-t)*t^2,t=0..infinity);
```

6. Graphing

(1) A basic plot: Plot $3x^2 - 8, -5 \leq x \leq 5$

```
plot(3*x^2-8,x=-5..5);
plot(3*x^2-8,x=-5..5,y=-3..10);
```

Be careful about the window size!

```
plot(x^3+1-exp(x),x=-8..8);
plot(x^3+1-exp(x),x=-8..8,y=-5..15);
```

(2) Plot two curves together, and add some color. The second command is useful when the printer does not do color:

```
plot( [cos(x),x^2],x=-4..4,color=[blue,black]);
plot( [cos(x),x^2],x=-4..4,color=[blue,black],linestyle=[1,4]);
```

(3) Plot points and line segments:

```
plot([[3,2],[-2,3],[2,-1]],style=point,color=blue,symbol=circle);
plot([[3,2],[-2,3],[2,-1]],style=line, color=red);
```

(4) Overlay several graphs on top of each other: Note the use of the colon

```
with(plots):
Fig1:=plot([3*x+5,9-x^2],x=-3..5,color=[green,red]):
Fig2:=plot([[-1,8],[4,-7]],style=point,color=blue,symbol=circle):
display({Fig1,Fig2});
```

The last line could also read: `display([Fig1,Fig2]);`

(5) Parametric plots. Suppose

$$x(t) = \cos(3t), \quad y(t) = \sin(t + sin(5t))$$

then we can plot the curve as:

```
x:=cos(3*t);
y:=sin(t+sin(5*t));
plot( [x,y,t=1..10] );
```